Intrinisically Typed Syntax That Works (Fresh Perspective)

ANONYMOUS AUTHOR(S)

Intrinsically typed syntax promises elegant mechanizations: well-typed terms are represented by well-typed data, substitution preserves typing by construction, and type safety proofs collapse to straightforward induction. For simple calculi such as the simply-typed lambda calculus this promise is fulfilled. For realistic systems, however, the technique often becomes impractical. In dependently typed proof assistants, intrinsically typed encodings of polymorphic calculi lead to pervasive type-changing equalities: renamings and substitutions indexed by other substitutions force proofs to transport terms across equal types that are not definitionally equal. This phenomenon—informally known as transfer hell—causes even routine metatheoretic arguments to be dominated by administrative reasoning.

This fresh perspective shows how to make intrinsically typed syntax work in practice for polymorphic calculi. We give extacts of a mechanization of the metatheory of System F and extend it to System Fω in Agda while avoiding explicit reasoning about transports almost entirely. The key is to internalize the equational theory of the σ-calculus for substitutions and make it definitional. We rely on proved σ-calculus laws for a functional representation of substitutions and install them into Agda's rewrite mechanism, obtaining canonical behavior for substitution and renaming without explicit transports. The main technical challenge is ensuring that the resulting rewrite system remains terminating and confluent together with Agda's built-in reductions.

The use of rewriting dramatically simplifies proofs of metatheoretic properties: typing preservation becomes definitional, substitution lemmas disappear, and even well-behaved type conversions can be handled intrinsically. Our experience suggests a general methodology: when intrinsically typed syntax generates transports, the correct solution is often not more lemmas but better definitional equality. We propose some practical guidelines towards this methodology.

ACM Reference Format:
Anonymous Author(s). 2018. Intrinisically Typed Syntax That Works (Fresh Perspective). *ACM Trans. Program. Lang. Syst.* 37, 4, Article 111 (August 2018), 17 pages. https://doi.org/XXXXXXX.XXXXXXX

1 Introduction

Intrinsically typed syntax is widely recommended for mechanizing type systems in dependently typed proof assistants. By representing well-typed terms as well-typed data, typing invariants hold by construction as in a Church-style presentation, i.e., making typing preservation trivial. For small calculi this approach works beautifully. However, for realistic polymorphic calculi such as System F, mechanizations often become complicated: conceptually simple proofs become cluttered with transports across equal types and administrative equalities.

The simply-typed lambda calculus is the prime example where intrinsically typed syntax works like a charm. Figure 1 shows typical definitions of the syntax of types, type contexts, variables, and expressions, which are indexed by contexts and types (left column). Building on these definitions it is straightforward to define a small-step operational semantics and proving type safety. This is because subject reduction holds by construction and progress follows by a simple inductive proof (right column). The style of these definitions is adapted from the textbook by Kokke et al. [7].

ACM 1558-4593/2018/8-ART111
https://doi.org/XXXXXXX.XXXXXXX

```
data Type : Set where
  𝟙    : Type
  _⇒_ : Type → Type → Type

data Ctx : Set where
  ∅    : Ctx
  _▷_ : Ctx → Type → Ctx

data _∋_ : Ctx → Type → Set where
  here  : (Γ ▷ T) ∋ T
  there : Γ ∋ T → (Γ ▷ U) ∋ T

data _⊢_ Γ : Type → Set where
  con : Γ ⊢ 𝟙
  var : Γ ∋ T → Γ ⊢ T
  lam : (Γ ▷ T) ⊢ U → Γ ⊢ (T ⇒ U)
  app : Γ ⊢ (T ⇒ U) → Γ ⊢ T → Γ ⊢ U

data _⟶_ {Γ} {T} : Γ ⊢ T → Γ ⊢ T → Set where
  β-lam : app {Γ = Γ} (lam e₁) e₂ ⟶ (e₁ [ e₂ ])
  ξ-app : e₁ ⟶ e₁′ → app e₁ e₂ ⟶ app e₁′ e₂

data Value {Γ} : Γ ⊢ T → Set where
  con : Value con
  lam : (e : (Γ ▷ T) ⊢ U) → Value (lam e)

data Progress : Γ ⊢ T → Set where
  done : Value e → Progress e
  step : e ⟶ e′ → Progress e

progress : (e : ∅ ⊢ T) → Progress e
progress con       = done con
progress (lam e)   = done (lam e)
progress (app e₁ e₂)
  with progress e₁
... | step x       = step (ξ-app x)
... | done (lam e) = step β-lam
```

Fig. 1. Call-by-name simply-typed lambda calculus

We elide the definition of the substitution operator _[_], as we discuss substitutions in detail in Section 2.2.

The simply-typed lambda calculus therefore represents the ideal scenario: intrinsic typing elides type preservation proofs, thus making type safety proofs so concise that they can serve as textbook examples [7]. Nothing comparable to transport reasoning arises in this development as only definitional properties of substitutions are required. The remainder of this paper explains why the same approach breaks down for polymorphic calculi and how to recover some of its simplicity.

Transfer hell is the colloquial term for being forced to apply type-changing lemmas to terms that are already known to be equal. The root problem is that substitutions that are propositionally equal are not definitionally equal, so terms must repeatedly be transported across equal types. This situation is annoying because proof assistants like Agda or Coq plainly refuse to let users even state this equality.[1] This phenomenon appears when substitutions themselves carry dependent typing information. In System F and Fω, expression substitutions depend on type substitutions and type substitutions depend on renamings; composing them yields types that are equal only propositionally, forcing transports throughout proofs. One workaround is to state definitions and proofs with explicit equivalence relations, but this practice is cumbersome, inefficient, and far away from the holy grail of definitional equality, which just works.

To our knowledge, previous intrinsically typed mechanization of System Fω (e.g., [3]) follow a natural design: expression syntax is indexed by typing derivations and substitutions are indexed by substitutions at the type level. While elegant at the level of definitions, this design makes proofs brittle. Because expression substitutions depend on type substitutions, and type substitutions depend on renamings, which again depend on type renamings, routine equalities require repeated transport across equal types, leading to directly to transfer hell.

[1] A situation that can be addressed using heterogeneous equality introduced in Conor McBride's thesis [8]. Benton et al. [2] discuss this approach in the context of an adequacy proof for a logical relation.

The message of this pearl is simple: when intrinsically typed syntax becomes complicated, the problem is often not the encoding of terms but the definition of equality on substitutions. Strengthening definitional equality can simplify metatheory more effectively than adding automation or additional lemmas.

In this paper, we examine the transfer hell arising by intrinsically-typed syntax encodings of System F and System Fω. Specifically, we consider the problems arising with type-renaming-indexed renamings and type-substitution-indexed substitutions. Our solution is to strengthen definitional equality rather than to add more lemmas. We make substitution compute canonically by internalizing the equational theory of substitutions (the σ-calculus) as definitional equality using Agda's rewriting mechanism. As a consequence, extensionally equal substitutions normalize to the same form and transports disappear from routine proofs. This approach is inspired by the tactics of Autosubst [10–12], but internalized by rewriting [4]. We build on a standard functional representation of renamings and substitutions [2], prove the lemmas underlying the equational laws of the σ-calculus, and then inject these equations into Agda's rewrite engine, i.e., make the σ-calculus equations definitional. The technical challenge is to ensure that the added equations remain terminating and confluent together with the proof assistant's built-in definitional equality.

We first exercise this program for plain System F in Section 2, significantly simplifying the transfer-ridden approach used by Saffrich et al. [9], and then apply the same techniques to System Fω in Section 5. In the latter, the main difference to previous work [3] lies in our intrinsic treatment of type conversion, which leads to significant simplifications. In particular, in our developments, transport lemmas disappear, subject reduction holds by construction, and type conversion in System Fω is handled intrinsically.

In summary, intrinsically typed syntax fails for polymorphic calculi because substitutions lack canonical equality; making them compute canonically restores the original promise of intrinsic typing.

Contributions

This functional pearl demonstrates how to mechanize polymorphic calculi using intrinsically typed syntax without pervasive transport reasoning.

- We identify the fundamental cause of "transfer hell" in intrinsically typed encodings of System F and System Fω: renamings and substitutions indexed by other substitutions produce equal types that are not definitionally equal.
- We show how to eliminate these transports by installing the equational laws of the σ-calculus for substitutions as definitional equality using Agda's rewriting mechanism, taking care to make the equations compatible with the built-in definitional equality.
- We demonstrate that the resulting rewrite system is compatible with Agda's definitional equality (in particular, terminating and confluent) for the functional representation of substitutions and verify that the resulting rewrite system is terminating and confluent with respect to the proof assistant's definitional equality.
- We mechanize parts of the metatheory of System F and extend it to System Fω, obtaining substantially simpler proofs and an intrinsic treatment of type conversion for the latter.
- From this experience we extract practical guidelines for designing intrinsically typed syntax in dependently typed proof assistants.

```
data Type (n : Nat) : Set where
  ‘_    : Fin n → Type n
  ∀α    : Type (suc n) → Type n
  _⇒_ : Type n → Type n → Type n
```

```
_ : Type 0 - a closed type:  ∀α. α→α
_ = ∀α (‘ zero ⇒ ‘ zero)
_ : Type 0 - ∀αβ. α→β→α
_ = ∀α (∀α (‘ suc zero ⇒ ‘ zero ⇒ ‘ suc zero))
```

Fig. 2. Syntax of System F types (Agda definition left, examples right)

2 Types for System F

2.1 Syntax

The syntax of System F types uses the standard de Bruijn encoding of variables by numbers. The type Type n (Figure 2) stands for the set of types with n type variables in scope. Henceforth, we call such n a *scope*. Variables for scope n are drawn from the type Fin n, i.e., the set $\{0, \ldots, n-1\}$, written as zero, suc zero, suc (suc zero), ... The $\forall\alpha$-binder introduces a new type variable, and $\Rightarrow$ is the infix function type constructor. We let T range over types.

2.2 Renaming and Substitution

The most important operations on types are consistent renaming of type variables and substitution. Figure 3 contains the first step in defining renamings (left column) and substitutions (right column). A type renaming ζ maps variables from one scope n_1 to another n_2. We write the type of type renamings as $n_1 \rightarrow^R n_2$. The definitions comprise (in order) weakening (wk^R), the identity renaming (id^R), extending a renaming with variable α ($\alpha \cdot^R \zeta$), applying a renaming to a variable ($\alpha \,\&^R\, \zeta$), composing two renamings ($\zeta_1 \fatsemi^R \zeta_2$), lifting a renaming (to move under a binder) ($\zeta \uparrow^R$), and applying a renaming to a type ($\mathsf{T}\,[\,\zeta\,]^R$).

A this point, the astute reader may wonder about the use of opaque in the definitions. The function symbols defined in an opaque block are usable from outside, but their computational behavior is hidden. Thus, the only normal definition is $\uparrow^R$ for lifting a renaming. This selective hiding is essential for our approach because we want to establish new definitional equalities (i.e., the laws of the σ-calculus with first-class renamigns) on top of the thus defined symbols. These new equalities interfere in a bad way with the standard definitional equalities, thus we hide the latter. To prove these equalities, we open the opaque definitions locally and proceed as usual.

For substitutions (right column of Figure 3) we proceed analogously. A type substitution η maps variables from scope n_1 to types defined over scope n_2. We write the type of type substitutions as $n_1 \rightarrow^S n_2$. The definitions (in order) comprise lifting a renaming to a substitution, extending a substitution, applying a substitution to variable, lifting a substitution (to move under a binder), applying a substitution to a type, and composing two substitutions.

2.3 Laws for Renaming and Substitution

Figure 4 gives the equational laws of the σ-calculus with first-class renamings [10]. All of these equalities are supported by proofs using the definitions from Figure 3, but the proofs are not included because they are standard.

The first group presents computation rules for substitutions starting with the application of a substitution to a variable. There is one equation for each possible form of a substitution: extended substitution, lifted renaming, and composition of substitutions. The final equation in this group characterizes lifting in terms of extension, weakening, and composition.

The second group contains laws about composition of substitutions. They describe how composition interacts with the other substitution primitives.

```
-- renamings
_→ᴿ_ : Nat → Nat → Set
n₁ →ᴿ n₂ = Fin n₁ → Fin n₂

opaque
  -- weakening
  wkᴿ : n →ᴿ suc n
  wkᴿ = suc

  -- identity renaming
  idᴿ : n →ᴿ n
  idᴿ α = α

  -- extend with new variable
  _•ᴿ_ : Fin n₂ → n₁ →ᴿ n₂ → suc n₁ →ᴿ n₂
  (α •ᴿ ζ) zero = α
  (_ •ᴿ ζ) (suc α) = ζ α

  -- apply renaming to variable
  _&ᴿ_ : Fin n₁ → n₁ →ᴿ n₂ → Fin n₂
  α &ᴿ ζ = ζ α

  -- left-to-right composition
  _⨟ᴿ_ : n₁ →ᴿ n₂ → n₂ →ᴿ n₃ → n₁ →ᴿ n₃
  (ζ₁ ⨟ᴿ ζ₂) α = ζ₂ (ζ₁ α)

-- lifting
_↑ᴿ : n₁ →ᴿ n₂ → suc n₁ →ᴿ suc n₂
_↑ᴿ ζ = zero •ᴿ (ζ ⨟ᴿ wkᴿ)

-- apply renaming to type
_[_]ᴿ : Type n₁ → n₁ →ᴿ n₂ → Type n₂
(` α) [ ζ ]ᴿ = ` (α &ᴿ ζ)
(∀α T) [ ζ ]ᴿ = ∀α (T [ ζ ↑ᴿ ]ᴿ)
(T₁ ⇒ T₂) [ ζ ]ᴿ = (T₁ [ ζ ]ᴿ) ⇒ (T₂ [ ζ ]ᴿ)
```

```
-- substitutions
_→ˢ_ : Nat → Nat → Set
n₁ →ˢ n₂ = Fin n₁ → Type n₂

opaque
  -- lift renaming to substitution
  ⟨_⟩ : n₁ →ᴿ n₂ → n₁ →ˢ n₂
  ⟨ ζ ⟩ α = ` (α &ᴿ ζ)

  -- extend with new type
  _•ˢ_ : Type n₂ → n₁ →ˢ n₂ → suc n₁ →ˢ n₂
  (T •ˢ η) zero = T
  (T •ˢ η) (suc α) = η α

  -- apply substitution to variable
  _&ˢ_ : Fin n₁ → n₁ →ˢ n₂ → Type n₂
  α &ˢ η = η α

  -- lifting
  _↑ˢ : n₁ →ˢ n₂ → suc n₁ →ˢ suc n₂
  _↑ˢ η = (` zero) •ˢ λ α → (η α) [ wkᴿ ]ᴿ

-- apply substitution to type
_[_]ˢ : Type n₁ → n₁ →ˢ n₂ → Type n₂
(` α) [ η ]ˢ = α &ˢ η
(∀α T) [ η ]ˢ = ∀α (T [ η ↑ˢ ]ˢ)
(T₁ ⇒ T₂) [ η ]ˢ = (T₁ [ η ]ˢ) ⇒ (T₂ [ η ]ˢ)

opaque
  -- left-to-right composition
  _⨟ˢ_ : n₁ →ˢ n₂ → n₂ →ˢ n₃ → n₁ →ˢ n₃
  (η₁ ⨟ˢ η₂) α = (η₁ α) [ η₂ ]ˢ
```

Fig. 3. Renamings and substitutions on types

The analogous rules for renamings from the first two groups were omitted for space reasons.

The third group presents laws that govern the composition of different kinds of term traversal. All four combinations of applying renamings and substitutions to a type are covered.

The final group contains covers laws that mediate between substitutions and renamings. The extraction of renamings from substitutions in general requires a dedicated solving strategy [12] and we have omitted some additionally required coincidence laws from the figure.

Orienting these equations from left to right results in a confluent rewrite system that we install in Agda using its REWRITE pragma [4]. From now on, all equations in Figure 4 are definitional!

As a sanity check, we show that lifting a substitution and the action of a substitution satisfy the functor laws. All of these lemmas hold definitionally, some correspond directly to equations from

$$
\begin{array}{lll}
\text{-- computing substitutions} & & \\
\text{beta-ext-zero} & : \mathsf{zero}\ \&^S\ (T \cdot^S \eta) & \equiv T \\
\text{beta-ext-suc} & : \mathsf{suc}\ \alpha\ \&^S\ (T \cdot^S \eta) & \equiv \alpha\ \&^S\ \eta \\
\text{beta-rename} & : \alpha\ \&^S\ \langle\ \zeta\ \rangle & \equiv \text{`}\ (\alpha\ \&^R\ \zeta) \\
\text{beta-comp} & : \alpha\ \&^S\ (\eta_1 \fatsemi^S \eta_2) & \equiv (\alpha\ \&^S\ \eta_1)\ [\ \eta_2\ ]^S \\
\text{beta-lift} & : \eta \uparrow^S & \equiv (\text{`}\ \mathsf{zero}) \cdot^S (\eta \fatsemi^S \langle\ \mathsf{wk}^R\ \rangle) \\
\text{-- interaction between substitutions} & & \\
\text{associativity} & : (\eta_1 \fatsemi^S \eta_2) \fatsemi^S \eta_3 \equiv \eta_1 \fatsemi^S (\eta_2 \fatsemi^S \eta_3) & \\
\text{distributivity} & : (T \cdot^S \eta_1) \fatsemi^S \eta_2 & \equiv (T\ [\ \eta_2\ ]^S) \cdot^S (\eta_1 \fatsemi^S \eta_2) \\
\text{interact} & : \langle\ \mathsf{wk}^R\ \rangle \fatsemi^S (T \cdot^S \eta) & \equiv \eta \\
\text{comp-id}_r & : \eta \fatsemi^S \langle\ \mathsf{id}^R\ \rangle & \equiv \eta \\
\text{comp-id}_l & : \langle\ \mathsf{id}^R\ \rangle \fatsemi^S \eta & \equiv \eta \\
\eta\text{-id} & : (\text{`}\ \mathsf{zero}\ \{n\}) \cdot^S \langle\ \mathsf{wk}^R\ \rangle \equiv \langle\ \mathsf{id}^R\ \rangle & \\
\eta\text{-law} & : (\mathsf{zero}\ \&^S\ \eta) \cdot^S (\langle\ \mathsf{wk}^R\ \rangle \fatsemi^S \eta) \equiv \eta &
\end{array}
$$

$$
\begin{array}{ll}
\text{-- composing renamings and substitutions} & \\
\text{identity}_r & : T\ [\ \mathsf{id}^R\ ]^R \equiv T \\
\text{compositionality}^{RS} & : (T\ [\ \zeta_1\ ]^R)\ [\ \eta_2\ ]^S \equiv T\ [\ \langle\ \zeta_1\ \rangle \fatsemi^S \eta_2\ ]^S \\
\text{compositionality}^{RR} & : (T\ [\ \zeta_1\ ]^R)\ [\ \zeta_2\ ]^R \equiv T\ [\ \zeta_1 \fatsemi^R \zeta_2\ ]^R \\
\text{compositionality}^{SR} & : (T\ [\ \eta_1\ ]^S)\ [\ \zeta_2\ ]^R \equiv T\ [\ \eta_1 \fatsemi^S \langle\ \zeta_2\ \rangle\ ]^S \\
\text{compositionality}^{SS} & : (T\ [\ \eta_1\ ]^S)\ [\ \eta_2\ ]^S \equiv T\ [\ \eta_1 \fatsemi^S \eta_2\ ]^S \\
\text{-- transforming substitutions to renamings} & \\
\text{coincidence} & : T\ [\ \langle\ \zeta\ \rangle\ ]^S \equiv T\ [\ \zeta\ ]^R \\
\text{coincidence-comp} & : \langle\ \zeta_1\ \rangle \fatsemi^S \langle\ \zeta_2\ \rangle \equiv \langle\ \zeta_1 \fatsemi^R \zeta_2\ \rangle
\end{array}
$$

Fig. 4. Equational laws of the σ-calculus

the σ- calculus in Figure 4. The latter are marked with $*$.[2] The lifting laws are shown on the left, the application laws on the right. The analogous results for renamings follow exactly the same pattern.

$$
\begin{array}{ll}
\text{lifts*-id} : (\mathsf{id}^S\ \{n\} \uparrow^S) \equiv \mathsf{id}^S & \text{sub*-id} : T\ [\ \mathsf{id}^S\ ]^S \equiv T \\
\text{lifts*-id} = \mathsf{refl} & \text{sub*-id} = \mathsf{refl} \\
\text{lifts*-comp} : (\eta' \fatsemi^S \eta) \uparrow^S \equiv (\eta' \uparrow^S) \fatsemi^S (\eta \uparrow^S) & \text{sub*-var} : (\text{`}\ \alpha)\ [\ \eta\ ]^S \equiv \alpha\ \&^S\ \eta \\
\text{lifts*-comp} = \mathsf{refl} & \text{sub*-var} = \mathsf{refl} \quad - * \\
 & \text{sub*-comp} : T\ [\ \eta \fatsemi^S \eta'\ ]^S \equiv (T\ [\ \eta\ ]^S)\ [\ \eta'\ ]^S \\
 & \text{sub*-comp} = \mathsf{refl} - *
\end{array}
$$

2.4 A Note on Safety

Turning propositional equalities into definitional ones is a delicate act in a proof assistant based on a rich dependent type theory, where we care a lot about safety and consistency. Systems like Dedukti [] are more centered on rewrite rules, but do not feature the full computational machinery

[2]The naming of the lemmas is taken from Chapman et al. [3] to enable direct comparisons.

that Agda offers. Combining rewrite rules dependent type theories, such as MLTT [] or CiC [] is a recently emerging field, first explored by the addition of rewrite rules to Agda and Rocq [], and continued by ongoing research on *local* rewrite rules [].

Rewrite rules and the reductions of a dependently typed system interact in non-trivial ways, so having the right notion of when it is safe to add them is important. We follow [] to justify making the equations from Figure 4 definitional and recall their reasoning for the safety of the proposed Rewriting Type Theory (RTT), which has been successfully implemented in Agda. In RTT, the reduction relation is extended by turning propositional equalities into reduction rules: a term matched by the left-hand side of an equation reduces to the right-hand side, exactly what have done in the previous section. The two properties we wish to preserve when adding rewrite rules are *soundness* (i.e., subject reduction) and *consistency*. In particular, subject reduction is an implicit assumption of the type checker that we need to take care to uphold.

To uphold subject reduction, the critical property of the rewriting equations is that they are confluent *with respect to the already existing reduction rules*. Agda already provides a plethora of reduction rules: β, η (function and record expansion), δ (definition unfolding), and ι (dependent pattern matching clauses). The additionally rewritten rules must agree with all of them to be confluent. Agda checks this using a simple syntactic criterion called the triangle property. Confluence holds when every critical pair of left-hand sides can be joined in a single rewriting step that lies on a common reduct, This check does not require termination. Notably, termination is *not* a critical property, because it does not affect soundness or consistency, it is only observable through never-terminating type checking but does no further harm.

Some β and ι reductions arising from our functional definitions of renamings and substitutions in the previous section interfere with the intended rewrite equalities. This is why we explicitly hide these definitions inside an opaque block, so that only the equations we install with REWRITE drive their reduction behaviour.

Consistency is preserved as long as the equalities we rewrite are instantiated with proofs. This is easy to see, because RTT can be translated to extensional type theory [], a theory known to be consistent. The translation simply turns each rewrite rule reduction into an application of the *equality reflection* rule, which internalizes arbitrary propositional equalities as definitional ones, at the cost of decidability of type checking. Due to this lack of decidability, extensional type theory is not a common candidate for a basis of a proof assistant.

At present, Agda is the only mainstream proof assistant that supports the full feature set we require: functions whose reduction can be hidden inside opaque blocks, so they can be treated as symbols, instantiation of the rewrite equations over these symbols with proofs, and termination-independent confluence checking via the triangle property. Rocq currently lacks the ability to use defined functions as symbols inside rewrite equations and the ability to instantiate these equations with proofs, but aside from these differences our method transfers to Rocq directly using its own rewrite rule implementation [].

In our formalization, we enable the OPTIONS `--rewriting` together with `--confluence-check`, which makes the use of rewriting safe, provided we trust the implementation of the confluence checker. With these newly and safely gained definitional equalities in place, we can continue to define expressions and expression substitution without ending up in transfer hell.

3 Expressions for System F

3.1 Syntax

Working towards intrinsically-typed expressions, Figure 5 defines type weakening and single substitution along with type contexts, expression variables, and the syntax of expressions. Type

```
weaken : Type n → Type (suc n)
weaken T = T [ wk^R ]^R

_[_]* : Type (suc n) → Type n → Type n
T [ T′ ]* = T [ T′ ·^S id^S ]^S

data Ctx : Nat → Set where
  ∅   : Ctx zero
  _▷_ : Ctx n → Type n → Ctx n
  _▷* : Ctx n → Ctx (suc n)

data _∋_ : Ctx n → Type n → Set where
  zero : (Γ ▷ T) ∋ T
  suc  : Γ ∋ T → (Γ ▷ T′) ∋ T
  suc* : Γ ∋ T → (Γ ▷*) ∋ (weaken T)
```

```
data Expr : Ctx n → Type n → Set where
  `_  : Γ ∋ T →
        Expr Γ T
  λx  : Expr (Γ ▷ T₁) T₂ →
        Expr Γ (T₁ ⇒ T₂)
  _·_ : Expr Γ (T₁ ⇒ T₂) →
        Expr Γ T₁ →
        Expr Γ T₂
  Λα  : Expr (Γ ▷*) T →
        Expr Γ (∀α T)
  _·*_ : Expr Γ (∀α T) →
         (T′ : Type n) →
         Expr Γ (T [ T′ ]*)
```

Fig. 5. Type contexts, variables, and expressions

contexts and expression variables follow the work of Chapman et al. [3]. Type contexts Ctx keep track of expression variables and type variables. Correspondingly, variables are implemented as typed de Bruijn indices with an extra case suc* to skip over type variables in the context. Skipping requires weakening of the type as it is used under one more type binding.

Expressions are indexed by a context and a type. Each expression constructor corresponds to a typing rule in System F. The rule for type application relies on single substitution for types.

The definition of a small-step operational semantics requires expression renamings and substitution to implement beta reduction for value abstractions and type abstractions. As before, we aim to define a call-by-name semantics for concision.

3.2 Examples

As an example of System F expressions, we present some definitions for Church numerals. We start with their type, alongside the constant zero. For readability, we define aliases for type variables like `α for ` Z, `β for ` (S Z), and analogously for expression variables and binders.

```
ℕ^c : Type ∅ *
ℕ^c = ∀α ((`α ⇒ `α) ⇒ (`α ⇒ `α))

zero^c : Expr ∅ ℕ^c
zero^c = Λα (λg (λx `x))
```

We continue with the constant one and the Church numeral for two.

```
one^c : Expr ∅ ℕ^c
one^c = Λα (λg (λx (`g · `x)))

two^c : Expr ∅ ℕ^c
two^c = succ^c · (succ^c · zero^c)
```

The successor operation has a longer body and is shown full width.

```
succ^c : Expr ∅ (ℕ^c ⇒ ℕ^c)
succ^c = λx (Λα (λg (λx (`g · (((((` (suc (suc (suc* zero)))) ·* `α) · `g) · `x)))))
```

3.3 Renaming and Substitution

Expression substitutions are indexed by type substitutions.

```
_|_⇒ˢ_ : n₁ →ˢ n₂ → Ctx n₁ → Ctx n₂ → Set
η | Γ₁ ⇒ˢ Γ₂ = ∀ T → Γ₁ ∋ T → Expr Γ₂ (T [ η ]ˢ)
```

The identity substitution for expressions is straightforward to define. But this simplicity is due to the installed rewriting of type substitutions, which avoids an explicit application of a transport lemma using comp-id$_r$.

```
Idˢ : idˢ | Γ ⇒ˢ Γ
Idˢ _ = `_
```

Extending a substitution is standard.

```
_|_·ˢ_ : ∀ η → Expr Γ₂ (T [ η ]ˢ) → η | Γ₁ ⇒ˢ Γ₂ → η | (Γ₁ ▷ T) ⇒ˢ Γ₂
(_ | e ·ˢ σ) _ zero    = e
(_ | e ·ˢ σ) _ (suc x) = σ _ x
```

Due to the structure of contexts, we need two lifting lemmas, one for value bindings and another for type bindings. The former requires a transport lemma, which is applied by rewriting.

```
_|_⇑ˢ_ : ∀ η → η | Γ₁ ⇒ˢ Γ₂ → ∀ T → η | (Γ₁ ▷ T) ⇒ˢ (Γ₂ ▷ (T [ η ]ˢ))
η | σ ⇑ˢ T = η | (` zero) ·ˢ λ _ x → idᴿ | (σ _ x) [ Wkᴿ _ ]ᴿ
```

Lifting a substitution over a type binding is straightforward.

```
_|_↑ˢ* : ∀ η → η | Γ₁ ⇒ˢ Γ₂ → (η ↑ˢ) | (Γ₁ ▷*) ⇒ˢ (Γ₂ ▷*)
(η | σ ↑ˢ*) = (η ⨟ˢ ⟨ wkᴿ ⟩) | (` zero) ·ˢ* λ _ x → wkᴿ | (σ _ x) [ wkᴿ* ]ᴿ
```

These operations are sufficient to define the application of a substitution to an expression.

```
_|_[_]ˢ : (η : n₁ →ˢ n₂) → Expr Γ₁ T → η | Γ₁ ⇒ˢ Γ₂ → Expr Γ₂ (T [ η ]ˢ)
η | (` x) [ σ ]ˢ     = η | x &ˢ σ
η | (λx e) [ σ ]ˢ    = λx (η | e [ η | σ ⇑ˢ _ ]ˢ)
η | (Λα e) [ σ ]ˢ    = Λα ((η ↑ˢ) | e [ η | σ ↑ˢ* ]ˢ)
η | (e · e₁) [ σ ]ˢ  = (η | e [ σ ]ˢ) · (η | e₁ [ σ ]ˢ)
η | (e ·* T′) [ σ ]ˢ = (η | e [ σ ]ˢ) ·* (T′ [ η ]ˢ)
```

3.4 Semantics

We are now in position to define a small-step operational semantics. The definition of single substitution, needed for beta reduction, is analogous to the one on types: we build the substitution by extending the identity substitution with the argument. In addition, a type substitution into expression is needed for beta reduction of type applications. This substitution is also implemented by an expression substitution indexed by a type substitution that performs the actual change.

```
_[_] : Expr (Γ ▷ T′) T → Expr Γ T′ → Expr Γ T
e [ e′ ] = idˢ | e [ idˢ | e′ ·ˢ Idˢ ]ˢ
```

```
_[*_*] : Expr (Γ ▷*) T → (T′ : Type n) → Expr Γ (T [ T′ ]*)
e [* T′ *] = (T′ ·ˢ idˢ) | e [ idˢ | T′ ·ˢ* Idˢ ]ˢ
```

With substitution in place, the definition of call-by-name small-step reduction is straightforward. As usual, such a definition of reduction for intrinsically-typed syntax implies subject reduction by construction.

```
data _⟶_ : Expr Γ T → Expr Γ T → Set where
  β-λ : (λx e₁ · e₂) ⟶ (e₁ [ e₂ ])
```

```
data Value : Expr Γ T → Set where
  λx : (e : Expr (Γ ▷ T₁) T₂) → Value (λx e)
  Λα : Value e → Value (Λα e)

data Progress : Expr Γ T → Set where
  done : (v : Value e) → Progress e
  step : (e⟶e′ : e ⟶ e′) → Progress e

NoVar : Ctx n → Set
NoVar ∅       = ⊤
NoVar (Γ ▷ T) = ⊥
NoVar (Γ ▷*)  = NoVar Γ

noVar : NoVar Γ → ¬ (Γ ∋ T)
noVar nv (suc* x) = noVar nv x
```

```
progress : NoVar Γ → (e : Expr Γ T) → Progress e
progress nv (` x) = ⊥-elim (noVar nv x)
progress nv (λx e) = done (λx e)
progress nv (e · e₁)
  with progress nv e
... | done (λx e₂) = step β-λ
... | step e⟶e′ = step (ξ-·₁ e⟶e′)
progress nv (Λα e)
  with progress nv e
... | done v = done (Λα v)
... | step e⟶e′ = step (ξ-Λ e⟶e′)
progress nv (e ·* T′)
  with progress nv e
... | done (Λα v) = step β-Λ
... | step e⟶e′ = step (ξ-·₁* e⟶e′)
```

Fig. 6. Progress for System F

```
β-Λ : (Λα e ·* T′) ⟶ (e [* T′ *])
ξ-·₁  : e₁ ⟶ e₁′ → (e₁ · e₂) ⟶ (e₁′ · e₂)
ξ-·₁* : e ⟶ e′ → (e ·* T) ⟶ (e′ ·* T)
ξ-Λ : e ⟶ e′ → (Λα e) ⟶ (Λα e′)
```

To obtain type safety, it remains to prove progress. Figure 6 presents the complete proof of progress for call-by-name System F. A value is either a lambda or a big lambda where the body is already a value. An expression fulfills progress if it is either a value (done) or it can make a step (step). Due to the definition of value for big lambdas, we establish progress for type contexts that may contain type bindings, but no value bindings. These contexts are characterized by function NoVar. The actual proof of progress is by induction on the structure of expressions. The overall structure of this proof closely follows the corresponding proof by [3]. Their proof is more complicated because their semantics is call-by-value and their language supports isorecursive types.

4 Laws for Renaming and Substitution

Proving type safety for System F *requires* exercising the substitution theory at the *type level*. Without installing its laws as definitional equalities, many definitions become more cumbersome because transport lemmas have to be inserted to equalize the types of (Agda) expressions.

While proving type safety does *not* require exercising same theory at the *expression level*, it is still good practice to verify functorial and monadic properties of expression substitutions. Moreover, if we were to move on to constructing and proving properties of logical relations, then these properties would be needed [9].

As an example, we examine the composition of two expression substitutions. Its definition involves two type substitutions η_1 and η_2 that index the corresponding expression substitutions σ_1 and σ_2. The composed expression substitution is indexed by the composition of η_1 and η_2. Type checking this definition already exercises the laws for composition of type substitutions.

```
_,_|_⨟ˢ_ : ∀ η₁ η₂ → η₁ | Γ₁ ⇒ˢ Γ₂ → η₂ | Γ₂ ⇒ˢ Γ₃ → (η₁ ⨟ˢ η₂) | Γ₁ ⇒ˢ Γ₃
(_ , _ | σ₁ ⨟ˢ σ₂) _ x = _ | (σ₁ _ x) [ σ₂ ]ˢ
```

Here is the statement of functoriality of expression substitution: applying two substitutions one after the other to an expression is the same as applying their composition. This statement is a very prominent example of transfer hell: the types of the left and right side of the equality are different and would require a transfer without our rewriting machinery in place.

```
Compositionalityˢˢ : ∀ (e : Expr Γ₁ T) (η₁ : n₁ →ˢ n₂) (η₂ : n₂ →ˢ n₃) (σ₁ : η₁ | Γ₁ ⇒ˢ Γ₂) (σ₂ : η₂ | Γ₂ ⇒ˢ Γ₃)
  η₂ | (η₁ | e [ σ₁ ]ˢ) [ σ₂ ]ˢ ≡ (η₁ ⨟ˢ η₂) | e [ η₁ , η₂ | σ₁ ⨟ˢ σ₂ ]ˢ
```

We show the proof of this property just do demonstrate that type adjustments for type substitutions can be ignored and we can concentrate on the gist of the proof, rather than being overwhelmed by transfers. The remaining fly in the ointment is the explicit passing of the type substitutions η_1 and η_2. We chose to make them explicit; while Agda can infer them in many places, there are equally many places where inference is not possible.

```
Compositionalityˢˢ (` x)      η₁ η₂ σ₁ σ₂ = refl
Compositionalityˢˢ (λx {T₁ = T₁} e) η₁ η₂ σ₁ σ₂ = cong λx (begin
    η₂ | η₁ | e [ η₁ | σ₁ ⇑ˢ T₁ ]ˢ [ η₂ | σ₂ ⇑ˢ (T₁ [ η₁ ]ˢ) ]ˢ
  ≡⟨ Compositionalityˢˢ e η₁ η₂ (η₁ | σ₁ ⇑ˢ T₁) (η₂ | σ₂ ⇑ˢ (T₁ [ η₁ ]ˢ)) ⟩
    (η₁ ⨟ˢ η₂) | e [ η₁ , η₂ | η₁ | σ₁ ⇑ˢ T₁ ⨟ˢ (η₂ | σ₂ ⇑ˢ (T₁ [ η₁ ]ˢ)) ]ˢ
  ≡⟨ cong ((η₁ ⨟ˢ η₂) | e [_]ˢ) (Lift-Dist-Compˢˢ σ₁ σ₂) ⟩
    (η₁ ⨟ˢ η₂) | e [ (η₁ ⨟ˢ η₂) | η₁ , η₂ | σ₁ ⨟ˢ σ₂ ⇑ˢ T₁ ]ˢ
  □)
Compositionalityˢˢ (Λα e)   η₁ η₂ σ₁ σ₂ = cong Λα (begin
    (η₂ ↑ˢ) | (η₁ ↑ˢ) | e [ η₁ | σ₁ ↑ˢ* ]ˢ [ η₂ | σ₂ ↑ˢ* ]ˢ
  ≡⟨ Compositionalityˢˢ e (η₁ ↑ˢ) (η₂ ↑ˢ) (η₁ | σ₁ ↑ˢ*) (η₂ | σ₂ ↑ˢ*) ⟩
    ((η₁ ⨟ˢ η₂) ↑ˢ) | e [ (η₁ ↑ˢ) , η₂ ↑ˢ | η₁ | σ₁ ↑ˢ* ⨟ˢ (η₂ | σ₂ ↑ˢ*) ]ˢ
  ≡⟨ cong (((η₁ ⨟ˢ η₂) ↑ˢ) | e [_]ˢ) (lift*-dist-Compˢˢ η₁ η₂ σ₁ σ₂) ⟩
    ((η₁ ⨟ˢ η₂) ↑ˢ) | e [ (η₁ ⨟ˢ η₂) | η₁ , η₂ | σ₁ ⨟ˢ σ₂ ↑ˢ* ]ˢ
  □)
Compositionalityˢˢ (e₁ · e₂) η₁ η₂ σ₁ σ₂ = cong₂ _·_
  (Compositionalityˢˢ e₁ η₁ η₂ σ₁ σ₂)
  (Compositionalityˢˢ e₂ η₁ η₂ σ₁ σ₂)
Compositionalityˢˢ (e ·* T′) η₁ η₂ σ₁ σ₂ = cong (_·* (T′ [ η₁ ⨟ˢ η₂ ]ˢ))
  (Compositionalityˢˢ e η₁ η₂ σ₁ σ₂)
```

This proof relies on further properties, in particular Lift-Dist-Comp^{SS} (lifting a composition of substitutions is equal to composing their liftings) and lift*-dist-Comp^{SS} (its counterpart for lifting over type bindings), where the statement and proof would require explicit transfers. Essentially, these properties form a "food chain" that stops at the composition properties of renamings with renamings and liftings, all of which follow similarly. The supplemental material gives the full story.

4.1 Comparison to a Proof in Transfer Hell

We now contrast our development with the same proof carried out without installing the σ-calculus equations as definitional equalities. As a point of comparison, we take the supplement of the logical relation construction of Saffrich et al. [9], which also mechanizes the compositionality properties

for System F. To enable a direct comparison, we have aligned the surface syntax of their definitions with ours.

Inspecting the type of the compositionality lemma already exposes the core issue: because the σ-calculus equations are only propositional, the statement of the theorem must itself rely on subst to bridge equal-but-not-definitionally-equal indices.

```
Compositionality^SS-≡ :
  ∀ {η₁ : n₁ →^S n₂} {η₂ : n₂ →^S n₃}
    {Γ₁ : Ctx n₁} {Γ₂ : Ctx n₂} {Γ₃ : Ctx n₃}
    (e : Expr n₁ Γ₁ T)
    (σ₁ : η₁ | Γ₁ ⇒^S Γ₂) (σ₂ : η₂ | Γ₂ ⇒^S Γ₃) →
  let sub = subst (Expr n₃ Γ₃) (compositionality^SS T η₁ η₂) in
  sub (η₂ | (η₁ | e [ σ₁ ]^S) [ σ₂ ]^S) ≡ (η₁ ⨟^S η₂) | e [ η₁ , η₂ | σ₁ ⨟^S σ₂ ]^S
```

Every recursive call to the induction hypothesis introduces a fresh application of subst, which then has to be distributed to the outside of the term and combined with the substs arising from neighboring subterms before the goal can be closed. This bookkeeping is further complicated by the fact that the other appearances of type substitution lemmas required in the proof interact with the transported equalities in non-trivial ways. We illustrate the resulting complexity by zooming in on a single induction case—type application—of their compositionality lemma.

A first observation is that our development already saves one subst in the very definition of substitution application. Without the definitional σ-calculus, this saving is no longer available: the type application case requires the swap lemma of type $(T\,[\,T'\,]^*)\,[\,\eta\,] \equiv (T\,[\,\eta \uparrow^S\,])\,[\,T'\,[\,\eta\,]\,]^*$. to coerce the term to the expected type, leading to the following definition.

```
η | (_·*_ {T = T} e T') [ σ ]^S = subst (Expr _ _)
                                    (sym (swap^SS η T T'))
                                    ((η | e [ σ ]^S) ·* (T' [ η ]^S))
```

With this in place, we can examine the type application case of the compositionality proof itself. The authors employ heterogeneous equality [8] to avoid some of the administrative reasoning otherwise forced by transfer hell. Without heterogeneous equality, the proof would be considerably worse still, requiring several auxiliary lemmas that explicitly shuffle applications of subst around. Thus, the version below already constitutes one of the simplest forms of the argument.

```
let
  F₂ = Expr n₂ Γ₂ ; E₂ = sym (swap^SS η₁ T T')          ; sub₂ = subst F₂ E₂
  F₃ = Expr n₃ Γ₃ ; E₃ = sym (swap^SS (η₁ ⨟^S η₂) T T') ; sub₃ = subst F₃ E₃
  F₅ = Expr n₃ Γ₃ ; E₅ = sym (swap^SS η₂ (T [ η₁ ↑^S ]) (T' [ η₁ ]^S)) ; sub₅ = subst F₅ E₅
in
R.begin
  η₂ | (η₁ | (e ·* T') [ σ₁ ]^S) [ σ₂ ]^S
R.≅⟨ refl ⟩                                                    - (1)
  η₂ | (sub₂ ((η₁ | e [ σ₁ ]^S) ·* (T' [ η₁ ]^S))) [ σ₂ ]^S
R.≅⟨ H.cong₂ {B = Expr n₂ Γ₂} (λ _ ■ → η₂ | ■ [ σ₂ ]^S)        - (2)
       (H.≡-to-≅ (sym E₂))
       (H.≡-subst-removable F₂ E₂ _) ⟩
  η₂ | ((η₁ | e [ σ₁ ]^S) ·* (T' [ η₁ ]^S)) [ σ₂ ]^S
R.≅⟨ refl ⟩                                                    - (3)
```

```
  sub₅ ((η₂ | (η₁ | e [ σ₁ ]ˢ) [ σ₂ ]ˢ) ·* (T′ [ η₁ ]ˢ [ η₂ ]ˢ))
R.≅⟨ H.≡-subst-removable F₅ E₅ _ ⟩                                    - (4)
  (η₂ | (η₁ | e [ σ₁ ]ˢ) [ σ₂ ]ˢ) ·* (T′ [ η₁ ]ˢ [ η₂ ]ˢ)
R.≅⟨ Hcong₃ {B = λ ■ → Expr n₃ Γ₃ (∀α ■)} {C = λ _ _ → Type n₃ _ } - (5)
       (λ _ ■₁ ■₂ → ■₁ ·* ■₂)
       (H.≡-to-≅ (lift-compositionality^SS T η₁ η₂))
       (Compositionality^SS e σ₁ σ₂)
       (H.≡-to-≅ (compositionality^SS T′ η₁ η₂)) ⟩
  ((η₁ ⨟ˢ η₂) | e [ η₁ , η₂ | σ₁ ⨟ˢ σ₂ ]ˢ) ·* (T′ [ η₁ ⨟ˢ η₂ ]ˢ)
R.≅⟨ H.sym (H.≡-subst-removable F₃ E₃ _) ⟩                           - (6)
  sub₃ (((η₁ ⨟ˢ η₂) | e [ η₁ , η₂ | σ₁ ⨟ˢ σ₂ ]ˢ) ·* (T′ [ η₁ ⨟ˢ η₂ ]ˢ))
R.≅⟨ refl ⟩                                                          - (7)
  (η₁ ⨟ˢ η₂) | (e ·* T′) [ η₁ , η₂ | σ₁ ⨟ˢ σ₂ ]ˢ
R.□
```

The proof proceeds in seven steps.

(1) Unfold the definition of expression substitution at the inner application of σ_1, exposing the subst inside.
(2) Remove the subst that sits inside the term under consideration, justified by heterogeneous equality.
(3) Reveal yet another subst introduced by the substitution application of σ_2
(4) Remove the subst on the outside.
(5) Apply the induction hypothesis while simultaneously coercing the index according to the compositionality law for type substitutions.
(6) Reinsert the subst demanded by the proof goal on the outside.
(7) Fold the definition of expression substitution to conclude the proof.

Of these, steps (2), (4), the type coercion in (5), and (5), as well as the translation into heterogeneous equality, are entirely absent from our rewriting-based proof; only the application of the induction hypothesis and the implicit (un)folding of definitions remain.

We complement this qualitative comparison with a quantitative one. The following table records the lines of code and number of characters required to mechanize the compositionality law for expression substitutions, together with the cold type-checking time of the respective developments.

	With Rewriting		**Transfer Hell**	
	LOC	Chars	LOC	Chars
Type substitution + proofs	217	12 246	285	12 584
Expression substitution + proofs	199	11 465	1 405	69 846
Total	**416**	**23 711**	**1 690**	**82 430**
Type-check time (cold)	4.49 s		9.51 s	

As expected, the amount of code required to formalize type substitution is roughly the same in both developments. For expression substitutions, however, the picture changes: the transfer-hell version requires more than five times as many lines of code as the rewriting-based one. As a welcome side effect, the cold type-checking time is also reduced by roughly half when the σ-calculus equations are installed as rewrite rules.

4.2 Equational Laws for Renaming and Substitution

4.3 Practical Guidelines for Mechanizing Intrinsically Typed Substitutions

5 System Fω

System Fω extends System F with higher-kinded types, that is, abstraction over type functions. This calculus is the basis of Haskell's type system where type functions are limited to partial application of type constructors. Abstraction of higher-kinded types is key to defining type classes for functors, monads, etc. An extension of Fω with isorecursive types is the basis for Plutus, the language of the Cardano blockchain [3].

5.1 Definitions

We only give a glimpse of the definitions of the syntax of types and expressions, which are taken from [3] and hence generalize the definitions for System F from Section 2. We do not reiterate the development of renamings and substitutions, as it transfers mutatis mutandis: the typings get more involved, but the equational laws remain exactly the same.

For System Fω, type variables have more structure as in System F, as there are arrow kinds in addition to plain types. We use I, J, K to range over types.

```
data Kind : Set where
  ∗ : Kind
  _⇒_ : Kind → Kind → Kind
```

Consequently, kind contexts are not mere numbers, but lists of kinds that grow to the right. Kind contexts are ranged over by Φ, Ψ, Θ.

```
data Ctx* : Set where
  ∅ : Ctx*
  _▷*_ : Ctx* → Kind → Ctx*
```

Type variables have the same structure as plain variables in the simply-typed lambda calculus (see Figure 1).

```
data _∋*_ : Ctx* → Kind → Set where
  zero : (Φ ▷* K) ∋* K
  suc  : Φ ∋* K → (Φ ▷* J) ∋* K
```

The definition of the type syntax also mimicks the expressions of the simply-typed lambda calculus, but with an important twist, which deviates from the prior work.

As System Fω enjoys full beta reduction at the type level, we do not define application (with symbol \$) as a data constructor. Instead, we introduce it as a symbol and use rewriting to extend functions on the type syntax to include application.

As an example, we show the definition of applying a renaming to an expression. We elide the preceding definitions of lifting, extension, and the type substitution machinery because they are identical to the ones for System F in Section 2.2, save for the more elaborate type signatures.

```
_⨟_[_]ᴿ : (ζ : Φ →ᴿ Ψ) → Expr Γ₁ T → ζ ⨟ Γ₁ ⇒ᴿ Γ₂ → Expr Γ₂ (T [ ζ ]ᴿ)
_ ⨟ (` x) [ ρ ]ᴿ      = ` (ρ _ x)
_ ⨟ (λx e) [ ρ ]ᴿ     = λx (_ ⨟ e [ ρ ⇑ᴿ _ ]ᴿ)
_ ⨟ (Λα e) [ ρ ]ᴿ     = Λα (_ ⨟ e [ _ ⨟ ρ ↑ᴿ* _ ]ᴿ)
_ ⨟ (e₁ · e₂) [ ρ ]ᴿ  = (_ ⨟ e₁ [ ρ ]ᴿ) · (_ ⨟ e₂ [ ρ ]ᴿ)
```

```
ζ | (e ·* T) [ ρ ]R = (ζ | e [ ρ ]R) ·* (T [ ζ ]R)
ζ | conv e x [ ρ ]R = conv (ζ | e [ ρ ]R) (lemR x)
```

On top of the rewriting theory for type substitutions, we install type-level beta reduction as a rewrite rule, using single type substitution on the right side:

```
postulate
  β≡* : ∀ {Φ J K} {T₁ : Type (Φ ▷* J) K} {T₂ : Type Φ J} →
        _$_ {Φ}{J}{K} (λα {Φ}{J}{K} T₁) T₂ ≡ T₁ [ T₂ ]*

{-# REWRITE β≡* #-}
```

The combined theory is still confluent and terminating [1, 5].

Surprisingly, type-level beta reduction is not required to define the syntax of expressions, expression renamings and substitutions, defining the operational semantics, as well as the proof of type safety. Thanks to definitional beta conversion at the type level, the expression syntax does *not* require an explicit type conversion rule as it was the case in prior work [3]. Hence, the proof presented in Figure 6 for System F carries over, mutatis mutandis (i.e., adapting the type signatures).

In the next section, we develop some examples that genuinely rely on type conversion.

5.2 Examples

Chapman et al. [3] consider several examples to demonstrate their encoding of System Fωμ. Their first group of examples covers Church numerals in System F. These are covered in Section 3.2. Their second group of example covers recursive types. We have no counterpart for them. Unfortunately, they do not provide examples that exercise the full power of System Fω, namely higher-kinded types! We fill this gap in the following.

5.2.1 Church Numerals. First, we encode (simply-typed) Church numerals *in the type language*. As beta reduction on the type level is definitional in our development, normalization of Church numerals is performed by Agda's type checker.

The kind of type level Church numerals is as expected, and zero and successor are defined in the usual way. For readability, we define aliases like λβ and so on for the type level lambda λα. Analogously, we define aliases for type variables like ‘α for ‘ Z, ‘β for ‘ (S Z), and so on.

```
ℕᵏ : Kind
ℕᵏ = (∗ ⇒ ∗) ⇒ (∗ ⇒ ∗)

zeroᵏ : Type ∅ ℕᵏ
zeroᵏ = λβ (λα ‘α)

succᵏ : Type ∅ (ℕᵏ ⇒ ℕᵏ)
succᵏ = λγ (λβ (λα (‘β $ (‘γ $ ‘β) $ ‘α))))
```

We construct the Church numeral for 2 by applying the successor twice to zero (left). The proof that this term is equal to its normal form is definitional. We further define addition and construct 2 by adding 1 + 1 (right). The corresponding equality proof puts a significant load on the type checker and takes more than a minute to complete on the authors' machine.

```
two^k : Type ∅ ℕ^k
two^k = succ^k $ (succ^k $ zero^k)

_ : two^k ≡ λβ (λα (‘β $ (‘β $ ‘α)))
_ = refl

one^k : Type ∅ ℕ^k
one^k = λβ (λα (‘β $ ‘α))

add^k : Type ∅ (ℕ^k ⇒ (ℕ^k ⇒ ℕ^k))
add^k = λα (λα (λα (λα ((((‘ suc (suc (suc zero))) $ (‘ suc zero)) $ (((

_ : two^k ≡ (add^k $ one^k) $ one^k
_ = refl
```

5.2.2 Arrows. The final example is inspired by the work on arrows [6]. This work uses type classes to abstract over different instantiations of the function arrow. For clarity, we start with an Agda version of the function that we wish to implement in System Fω.

The function abst is a glorified identity function. It abstracts over a binary type constructor κ, two types β and α, then it takes an abstract application function *app* and applies it to an abstract function f and a suitable argument x. Function inst instantiates this function. It acts like an eliminator for abst. The function use puts both together.

```
ty : Set₁
ty = (κ : Set → Set → Set) (β α : Set) (app : κ β α → β → α) (f : κ β α) (x : β) → α

abst : ty
abst = λ κ β α → λ app f x → app f x

inst : ∀ α → ty → α → α
inst = λ α g x → g (λ β α → (β → α)) α α (λ f x → f x) (λ z → z) x

use : ∀ α → α → α
use = λ α x → inst α abst x
```

We start by defining the type in our encoding, then encode the defining expression equivalent to abst.

```
ty-enc : Type Φ ∗
ty-enc = ∀κ (∀β (∀α ((((‘κ $ ‘β) $ ‘α) ⇒ (‘β ⇒ ‘α))
                ⇒ (((‘κ $ ‘β) $ ‘α) ⇒ (‘β ⇒ ‘α)))))

abst-enc : Expr Γ ty-enc
abst-enc = Λκ (Λβ (Λα (λf (λy (λx ((‘f · ‘y) · ‘x))))))
```

To use this function, we encode functions inst and use.

```
inst-enc : Expr Γ (∀α (ty-enc ⇒ (‘α ⇒ ‘α)))
inst-enc = Λα (λg (λx ((((((‘g ·* (λβ (λα (‘β ⇒ ‘α)))) ·* ‘α) ·* ‘α)
                          · (λy (λx (‘y · ‘x))))
                          · λx ‘x)
                          · ‘x)))

use-enc : Expr ∅ (∀α (‘α ⇒ ‘α))
use-enc = Λα (λx (((inst-enc ·* ‘α) · abst-enc
```

The definitions of the encodings require beta conversion at the type level. These definitions are not accepted without the rewriting of beta reduction $\beta\equiv^*$.

6 Conclusion

Intrinsically typed syntax promises elegant mechanizations by making well-typed programs representable only as well-typed data, but for polymorphic calculi this promise is often undermined

by pervasive transport reasoning. Our experience with System F and System Fω shows that the difficulty does not stem from binding structure or indexing itself, but from a mismatch between substitution and definitional equality: substitutions that are extensionally equal fail to compute to a canonical form. By internalizing the equational laws of the σ-calculus as definitional equality, substitutions become canonical and routine transports disappear from proofs. The resulting developments recover the simplicity seen in the simply-typed case while scaling to higher-order polymorphism and intrinsic type conversion. More broadly, this pearl suggests a practical guideline for mechanizing type systems in dependently typed proof assistants: when intrinsically typed encodings become proof-heavy, a promising strategy is often to strengthen definitional equality rather than introduce additional lemmas or automation.

References

[1] M. Abadi, L. Cardelli, P.-L. Curien, and J.-J. Levy. 1989. Explicit substitutions. In *Proceedings of the 17th ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages* (San Francisco, California, USA) *(POPL '90)*. Association for Computing Machinery, New York, NY, USA, 31–46. doi:10.1145/96709.96712

[2] Nick Benton, Chung-Kil Hur, Andrew Kennedy, and Conor McBride. 2012. Strongly Typed Term Representations in Coq. *J. Autom. Reason.* 49, 2 (2012), 141–159. doi:10.1007/S10817-011-9219-0

[3] James Chapman, Roman Kireev, Chad Nester, and Philip Wadler. 2019. System F in Agda, for Fun and Profit. In *Mathematics of Program Construction: 13th International Conference, MPC 2019, Porto, Portugal, October 7–9, 2019, Proceedings* (Porto, Portugal). Springer-Verlag, Berlin, Heidelberg, 255–297. doi:10.1007/978-3-030-33636-3_10

[4] Jesper Cockx, Nicolas Tabareau, and Théo Winterhalter. 2021. The Taming of the Rew: A Type Theory with Computational Assumptions. *Proc. ACM Program. Lang.* 5, POPL, Article 60 (Jan. 2021), 29 pages. doi:10.1145/3434341

[5] Pierre-Louis Curien, Thérèse Hardin, and Jean-Jacques Lévy. 1996. Confluence properties of weak and strong calculi of explicit substitutions. *J. ACM* 43, 2 (March 1996), 362–397. doi:10.1145/226643.226675

[6] John Hughes. 2000. Generalising Monads to Arrows. *Sci. Comput. Program.* 37, 1-3 (2000), 67–111. doi:10.1016/S0167-6423(99)00023-4

[7] Wen Kokke, Jeremy G. Siek, and Philip Wadler. 2020. Programming Language Foundations in Agda. *Sci. Comput. Program.* 194 (2020), 102440. doi:10.1016/J.SCICO.2020.102440

[8] Conor McBride. 2000. Elimination with a Motive. In *Types for Proofs and Programs, International Workshop, TYPES 2000, Durham, UK, December 8-12, 2000, Selected Papers (Lecture Notes in Computer Science, Vol. 2277)*, Paul Callaghan, Zhaohui Luo, James McKinna, and Robert Pollack (Eds.). Springer, 197–216. doi:10.1007/3-540-45842-5_13

[9] Hannes Saffrich, Peter Thiemann, and Marius Weidner. 2024. Intrinsically Typed Syntax, a Logical Relation, and the Scourge of the Transfer Lemma. In *Proceedings of the 9th ACM SIGPLAN International Workshop on Type-Driven Development* (Milan, Italy) *(TyDe 2024)*. Association for Computing Machinery, New York, NY, USA, 2–15. doi:10.1145/3678000.3678201

[10] Steven Schäfer, Tobias Tebbi, and Gert Smolka. 2015. Autosubst: Reasoning with de Bruijn Terms and Parallel Substitutions. In *Interactive Theorem Proving*, Christian Urban and Xingyuan Zhang (Eds.). Springer International Publishing, Cham, 359–374. https://www.ps.uni-saarland.de/Publications/documents/SchaeferEtAl_2015_Autosubst_-Reasoning.pdf

[11] Kathrin Stark. 2020. *Mechanising Syntax with Binders in Coq.* Ph. D. Dissertation. Saarland University, Saarbrücken, Germany. https://www.ps.uni-saarland.de/Publications/documents/Stark_2020_Mechanising.pdf Dissertation zur Erlangung des Grades des Doktors der Ingenieurwissenschaften (Dr.-Ing.).

[12] Kathrin Stark, Steven Schäfer, and Jonas Kaiser. 2019. Autosubst 2: reasoning with multi-sorted de Bruijn terms and vector substitutions *(CPP 2019)*. Association for Computing Machinery, New York, NY, USA, 166–180. doi:10.1145/3293880.3294101